
1. [bookmark: _Toc122434485][bookmark: _Toc128428298]Overview
This document lists all the essencial changes needed to correct problems in the TTCN implementation of eCall over IMS test case 11.3.2, which is part of the LTE_A_PRO test suite in iwd-TTCN3-B2022-09_D22wk49.
With these changes applied the test case can be demonstrated to run with one or more UEs (see section 4). Execution log file is  provided as evidence. 
Contact:	Shaun Harry
	shaun.harry@keysight.com 	
0. [bookmark: _Toc123808182][bookmark: _Toc128428299]Verification Test Summary 
Test Case: 	11.3.2
ATS Version:	iwd-TTCN3-B2022-09_D22wk49
[bookmark: _Hlk37919671]System Simulator used:	Keysight S8704A Protocol Conformance Toolset
UE used:	Qualcomm SA515M
Verification Status:	PASS

1. [bookmark: _Toc122434488][bookmark: _Toc295288959][bookmark: _Toc128428300]Corrections required
1. [bookmark: _Toc128428301]Change 1
	Function name
	f_TC_11_3_2_EUTRA()

	Reason for change
	UE is already detached at the end of the test body

	Summary of change
	Corrected postamble state

	TTCN module
	 IMS_ECall.ttcn

	MCC160 Comment
	



Before change
  function f_TC_11_3_2_EUTRA() runs on EUTRA_PTC
  { /* eCall Only mode / T3445 / eCall inactivity procedure / Removal of eCall only restriction after a call to URI for test service */

    f_EUTRA_NAS_Init(LTE_AllCellsOnSamePLMN, c1, -, HANDLE_UM_DATA, USE_BIG_GRANTS);
    
    // Create cell and send out system information
    // Specific SIB1 support for IMS emergency support and eCall support will be set in default functions
    f_EUTRA_CellConfig_Def(eutra_CellA);
    
    // Insert USIM for eCall only enabled
    f_UT_USIM_Insert (UT, "36.508 Table 4.9.3.5-2");
    
    // Switch cell on
    f_EUTRA_SetCellPower(eutra_CellA, tsc_ServingCellRS_EPRE);
    
    // The UE is in state Switched OFF (state 1)
    f_EUTRA_TestBody_Set(true);
    
    fl_TC_11_3_2_Body();
    
    // Postamble
    f_EUTRA_TestBody_Set(false);

    f_EUTRA_Postamble(eutra_CellA, E1_IDLE); // optional IMS Deregistration is handled in the default, before the UE detaches
  }

    

After change
  function f_TC_11_3_2_EUTRA() runs on EUTRA_PTC
  { /* eCall Only mode / T3445 / eCall inactivity procedure / Removal of eCall only restriction after a call to URI for test service */

    f_EUTRA_NAS_Init(LTE_AllCellsOnSamePLMN, c1, -, HANDLE_UM_DATA, USE_BIG_GRANTS);
    
    // Create cell and send out system information
    // Specific SIB1 support for IMS emergency support and eCall support will be set in default functions
    f_EUTRA_CellConfig_Def(eutra_CellA);
    
    // Insert USIM for eCall only enabled
    f_UT_USIM_Insert (UT, "36.508 Table 4.9.3.5-2");
    
    // Switch cell on
    f_EUTRA_SetCellPower(eutra_CellA, tsc_ServingCellRS_EPRE);
    
    // The UE is in state Switched OFF (state 1)
    f_EUTRA_TestBody_Set(true);
    
    fl_TC_11_3_2_Body();
    
    // Postamble
    f_EUTRA_TestBody_Set(false);

    f_EUTRA_Postamble(eutra_CellA, E4_DEREGISTERED); // optional IMS Deregistration is handled in the default, before the UE detaches
  }
   

1. [bookmark: _Toc128428302]Change 2
	Function name
	fl_TC_11_3_2_Body()

	Reason for change
	1. Proper implementation is required to receive the initial RRC connection request for an expected RRC connection establishment procedure.
2. Extra trigger needed to IMS PTC, just need to inform IMS PTC to perform ecall establishment.
3. Timer t_T3445Min timer is currently not started.
4. t_T3412 should be started following RRCConnectionRelease at step 67
5. T3412 value is format TV in TRACKING AREA UPDATE ACCEPT but currently template cs_GprsTimer_v is being used which does not include the IEI field.
6. Currently the TRACKING AREA UPDATE ACCEPT message sent from function f_EUTRA_TrackingAreaUpdate() does not indicate support of emergency bearer services, this can cause an eCall-Only device to detach immediately.
7. DETACH REQUEST at step 72 will be integrity protected but not ciphered.

	Summary of change
	1. Replaced by common function to await RRC Connection Request sent by the UE.
2. Added extra trigger to EUTRA PTC
3. Add starting of t_T3445Min
4. T3412 timer started after step 67.
5. Use template cs_GprsTimer_tv for T3412 in TRACKING AREA UPDATE ACCEPT message
6. Pass NW feature support into f_EUTRA_TrackingAreaUpdate() indicating support of emergency bearer services, see change 2.6
7. Correct security status for steps 72.

	TTCN module
	 IMS_ECall.ttcn

	MCC160 Comment
	




Before change
  function fl_TC_11_3_2_Body() runs on EUTRA_PTC
  {
    var GutiParameters_Type v_Guti1Params := f_EUTRA_CellInfo_GetGuti(eutra_CellA);
    var EUTRA_SecurityParams_Type v_SecurityParams := f_EUTRA_Security_Get();
    var ProcedureTransactionIdentifier v_EPS_TI  := '01'O;
    var APNandPCOs_Type v_APNandPCOs;
    var template (omit) ProtocolConfigOptions v_PcoFromUE := omit;
    var template (omit) ProtocolConfigOptions v_PcoToUE := omit;
    var PDN_Index_Type v_PdnIndex := f_EUTRA_InitialRegistrationPdnIndex(firstPDN);
    var template (value) GPRS_Timer v_T3412Value := cs_GprsTimer_v('010'B, '11111'B); //T3412 value = 186 (6 x 31) mins
    var template AdditionalUpdateType v_AdditionalUpdateType := f_GetAdditionalUpdateType ();
    var NAS_AttDetValue_Type v_DetachType;
    var NAS_MSG_Indication_Type v_NasInd;
    var boolean v_T3445Min_elapsed := false;
    var float v_EcallDeregisteredInactive := 60.0;
    var EUTRA_InitialRegistration_PdnStatus v_PdnStatus;
    var LinkedEpsBearerIdentity v_LinkedEpsBearerId := {idValue := hex2bit(tsc_EpsDefaultBearerId2ndPDN)};
    var SRB_COMMON_IND v_ReceivedAsp;
    var ProcedureTransactionIdentifier v_EpsTi;
    var B8_Type v_EsmCauseValue;

    timer t_T3412Max := f_EUTRA_SetTimerToleranceMax(eutra_CellA, nasTimer, 11160.0); // 186 minutes
    timer t_T3445Max := f_EUTRA_SetTimerToleranceMax(eutra_CellA, nasTimer, 43200.0); // 12 hours
    timer t_T3445Min := f_EUTRA_SetTimerToleranceMin(eutra_CellA, nasTimer, 43200.0); // 12 hours

    f_EUTRA_MobileInfo_SetNetworkFeatureSupport(cs_EPS_NwkFtSup_IMSVoice_EMCBS); // @sic R5s180648 sic@ emergency bearer services in S1 mode and IMS voice over PS session in S1 mode are supported

    //@siclog "Step 1" siclog@
    // The UE is switched on.
    f_EUTRA_SwitchOnUEandStartIP(eutra_CellA);

    //@siclog "Step 2" siclog@
    // Wait for 60s for the UE to enter EMM-DEREGISTERED.eCALL-INACTIVE state
    f_Delay(v_EcallDeregisteredInactive);

    //@siclog "Step 3" siclog@
    // An eCall to URI for test service is initiated by dialling the eCall Test Number in the EFFDNURI field of the USIM (tel:123456)
    f_UT_InitiateECall(UT, TEST);
      
    //@siclog "Step 3A" siclog@
    SRB.check(receive(car_SRB0_RrcPdu_IND(?, cr_RRCConnectionRequest(cr_EstablishmentCause_MoSignalling))));
    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 3A");

    //@siclog "Step 3B-4" siclog@
    // Check: Does the UE sends an ATTACH REQUEST message with a PDN CONNECTIVITY REQUEST message to request PDN connectivity to the default PDN. EPS attach type = "combined EPS/IMSI attach"?
    v_NasInd := f_EUTRA_RRCConnectionSetupComplete_Def(eutra_CellA,
                                                       cr_508_RRCConnectionSetupComplete(tsc_RRC_TI_Def, ?),
                                                       cr_NAS_IndicationWithPiggybacking((tsc_SHT_NoSecurityProtection, tsc_SHT_IntegrityProtected),
                                                                                         cr_508_ATTACH_REQUEST(tsc_EAttachType_Combined,
                                                                                                               v_AdditionalUpdateType),
                                                                                         cr_508_PDN_CONNECTIVITY_REQUEST(tsc_PdnRequest_InitialAttach,
                                                                                                                         f_GetPdnType())));

    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 4");
    
    v_EPS_TI := v_NasInd.Pdu.PiggybackedPduList[0].Msg.pDN_CONNECTIVITY_REQUEST.procedureTransactionIdentifier;
    v_SecurityParams.NAS_SecurityCap := f_EUTRA_NB_FillNAS_SecurityCapability(v_NasInd.Pdu.Msg.aTTACH_REQUEST);
    f_EUTRA_Security_Set ( v_SecurityParams );//Saving security parameters
   
    //@siclog "Step 5-13" siclog@
    // Steps 5-13 of the generic procedure for UE registration specified in TS 36.508 subclause 4.5.2.3 are performed.
    v_APNandPCOs := f_EUTRA_InitialRegistration_Step5_13(eutra_CellA, STATE2_IDLEUPDATE, v_NasInd);

    v_PcoFromUE := v_NasInd.Pdu.PiggybackedPduList[0].Msg.pDN_CONNECTIVITY_REQUEST.protocolConfigurationOptions;
    if (isvalue(v_APNandPCOs.protocolConfigurationOptions)) {
      v_PcoFromUE := v_APNandPCOs.protocolConfigurationOptions;
    }
    v_PcoToUE := f_GetDefaultProtocolConfigOptions (v_PcoFromUE, v_PdnIndex);


    //@siclog "Step 14-15" siclog@
    // The SS transmits an ATTACH ACCEPT message with the ACTIVATE DEFAULT EPS BEARER CONTEXT REQUEST message and SS assigns value of 186 minutes for the timer T3412.
    f_EUTRA_InitialRegistration_Step14_15(eutra_CellA,
                                          NORMAL,
                                          v_NasInd.Pdu.Msg.aTTACH_REQUEST.addUpdateType,
                                          v_EPS_TI,
                                          f_CheckPCOforIPallocationViaNas (v_PcoFromUE),
                                          v_PcoToUE,
                                          v_APNandPCOs.accessPointName,
                                          v_T3412Value);// 186 minutes

    //@siclog "Step 16-17A4" siclog@
    // Steps 16-16c1 of the generic procedure for UE registration specified in TS 36.508 subclause 4.5.2.3 are performed.
    v_PdnStatus := f_EUTRA_InitialRegistration_Step16_16c1 (eutra_CellA);
    
    //@siclog "Step 17A5a1-17A5b1" siclog@
    if (v_PdnStatus != singlePDN) {     // @sic R5-227567 sic@
      if (pc_UE_supports_user_initiated_PDN_disconnect) {
        // Step 1
        f_UT_DeactivateBearer(UT, hex2int(tsc_EpsDefaultBearerId2ndPDN));
      
        // Step 9: The UE transmits PDN DISCONNECT REQUEST
        SRB.receive(car_SRB2_NasPdu_IND(eutra_CellA, cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered, cr_PDNDisconnectReq(v_LinkedEpsBearerId)))) -> value v_ReceivedAsp;
        v_EpsTi := v_ReceivedAsp.Signalling.Nas[0].Pdu.Msg.pDN_DISCONNECT_REQUEST.procedureTransactionIdentifier;
        v_EsmCauseValue := tsc_ESM_Cause36_RegularDeactivation;
      } else {
        v_EpsTi := tsc_PTI_Unassigned;
        v_EsmCauseValue := tsc_ESM_Cause26_InsufficientResources;
      }
      // Steps 10-12
      f_EUTRA_DeactivateEPS_BearerContext(eutra_CellA, tsc_EpsDefaultBearerId2ndPDN, v_EpsTi, v_EsmCauseValue); // @sic R5-171372 sic@
      f_IP_Handling_StopPDN(IP, cs_IP_StopPDN(PDN_Internet));
    }

    //@siclog "Step 18-28 Void" siclog@
    
    // Wait until registration finished and Invite received
    f_IMS_IPCAN_WaitForTrigger(IMS[tsc_Index_IMS1]);
    //@siclog "Step 29-31" siclog@
    // The UE performs Steps 23-25 of the generic test procedure as described in TS 36.508 sub clause 4.5A.27.3
    f_EUTRA_ActivateDedicatedEpsBearer_Speech(eutra_CellA);
           
    //@siclog "Step 32-36" siclog@
    // Release eCall to URI for test service using the generic procedure described in TS 34.229-1 [35] sub clause C.33. @sic R5-196019 sic@
    //IMS call has been released.
    f_IMS_IPCAN_WaitForTrigger(IMS[tsc_Index_IMS1]);
    // The SS deactivates the EPS bearer. [This is the rest of C.33]
    f_EUTRA_DeactivateEPS_BearerContext (eutra_CellA,
                                         tsc_EpsDedicatedBearerId);
 
    //@siclog "Step 37" siclog@
    // The SS releases the RRC connection.
    f_EUTRA_RRC_ConnectionRelease(eutra_CellA);
    
    // start after the end of the eCall
    t_T3412Max.start;
    t_T3445Max.start;
    
    //@siclog "Step 38" siclog@
    // The SS waits 1 minute.
    f_Delay(60.0);

    //@siclog "Step 39-62" siclog@
    // Steps 1 to 24 of the generic test procedure for IMS MT speech call specified in TS 36.508 subclause 4.5A.7.3
    // from function f_EUTRA_GenericIMS_MT_SpeechCall(EUTRA_CellId_Type p_CellId) runs on EUTRA_PTC
    f_EUTRA_RbEst_Def(eutra_CellA); //Steps 1-8
    // Send trigger to IMS PTC
    f_IMS_IPCAN_SendCoOrdMsg(IMS[tsc_Index_IMS1]); // Steps 9 to 12 are executed in IMS PTC
    //Wait Trigger from IMS PTC: steps 1-4 have been done
    f_IMS_IPCAN_WaitForTrigger(IMS[tsc_Index_IMS1]);

    f_EUTRA_ActivateDedicatedEpsBearer_Speech(eutra_CellA, -, -, -, true);
    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 62");

    //@siclog "Step 63-66" siclog@
    // Generic test procedure for MT release of IMS call as described in annex C.33 of TS 34.229-1 [35] takes place.
    f_IMS_IPCAN_SendCoOrdMsg(IMS[tsc_Index_IMS1]);
    
    //Wait Trigger from IMS PTC: MT release has happened
    f_IMS_IPCAN_WaitForTrigger(IMS[tsc_Index_IMS1]);
    // The SS deactivates the EPS bearer. [This is the rest of C.33]
    f_EUTRA_DeactivateEPS_BearerContext (eutra_CellA,
                                         tsc_EpsDedicatedBearerId);
    
    //@siclog "Step 67" siclog@
    // The SS releases the RRC connection.
    f_EUTRA_RRC_ConnectionRelease(eutra_CellA);

    alt {
      //@siclog "Step 68" siclog@
      // Check: Does the UE transmit a TRACKING AREA UPDATE REQUEST message at the expiry of T3412 periodically for every 186 minutes and remains registered until T3445 expires?
      //@siclog "Step 69" siclog@
      // The SS transmits a TRACKING AREA UPDATE ACCEPT message.
      //@siclog "Step 70" siclog@
      // The UE sends TRACKING AREA UPDATE COMPLETE message.
      //@siclog "Step 71" siclog@
      // The SS releases the RRC connection.
      [] SRB.check(receive(car_SRB0_RrcPdu_IND(?, cr_RRCConnectionRequest(cr_EstablishmentCause_Any)))) {
          f_EUTRA_TrackingAreaUpdate(eutra_CellA, tsc_RRC_TI_Def, -, cr_EstablishmentCause_Any, tsc_EpsUpdate_Periodic, -, -, v_T3412Value);
          f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 68");
          t_T3412Max.start;
          repeat;
        }
      [] t_T3412Max.timeout {
          f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 68 - TAU expected was not received in time");
        }
      [] t_T3445Min.timeout {
          v_T3445Min_elapsed := true;
          t_T3412Max.stop;
        }
      }

    v_DetachType :=  f_GetEDetachType(NORMAL);
    alt {
      //@siclog "Step 72" siclog@
      // Check: Does the UE transmit a DETACH REQUEST message at expiry of T3444?
      [] SRB.check(receive(car_SRB0_RrcPdu_IND(?, cr_RRCConnectionRequest(cr_EstablishmentCause_Any)))) {
          v_NasInd := f_EUTRA_RRC_ConnEst_DefWithNas(eutra_CellA,
                                                     tsc_RRC_TI_Def,
                                                     ?,
                                                     (cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,
                                                                        cr_508_DETACH_REQUEST_MO(?,
                                                                                                 f_GutiParameters2MobileIdentity(omit, v_Guti1Params),
                                                                                                 cr_DetachType( '0'B, v_DetachType)))));
          t_T3445Max.stop;
          if  (v_T3445Min_elapsed) {
            f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 72");
          } else  {
            f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 72 - DETACH REQUEST expected was received too early");
          }
          //@siclog "Step 73" siclog@
          // The SS responds with DETACH ACCEPT message.
          SRB.send (cas_SRB1_NasPdu_REQ (eutra_CellA,
                                         cs_TimingInfo_Now,
                                         cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,
                                                        cs_508_DETACH_ACCEPT)));
        }
      [] t_T3445Max.timeout {
          f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 72 - DETACH REQUEST expected was not received");
        }
      }

    //@siclog "Step 74" siclog@
    // The SS releases the RRC connection.
    f_EUTRA_RRC_ConnectionRelease(eutra_CellA);
    
  }


After change
 function fl_TC_11_3_2_Body() runs on EUTRA_PTC
  {
    var GutiParameters_Type v_Guti1Params := f_EUTRA_CellInfo_GetGuti(eutra_CellA);
    var EUTRA_SecurityParams_Type v_SecurityParams := f_EUTRA_Security_Get();
    var ProcedureTransactionIdentifier v_EPS_TI  := '01'O;
    var APNandPCOs_Type v_APNandPCOs;
    var template (omit) ProtocolConfigOptions v_PcoFromUE := omit;
    var template (omit) ProtocolConfigOptions v_PcoToUE := omit;
    var PDN_Index_Type v_PdnIndex := f_EUTRA_InitialRegistrationPdnIndex(firstPDN);
    var template (value) GPRS_Timer v_T3412Value := cs_GprsTimer_v('010'B, '11111'B); //T3412 value = 186 (6 x 31) mins
    var template AdditionalUpdateType v_AdditionalUpdateType := f_GetAdditionalUpdateType ();
    var NAS_AttDetValue_Type v_DetachType;
    var NAS_MSG_Indication_Type v_NasInd;
    var boolean v_T3445Min_elapsed := false;
    var float v_EcallDeregisteredInactive := 60.0;
    var EUTRA_InitialRegistration_PdnStatus v_PdnStatus;
    var LinkedEpsBearerIdentity v_LinkedEpsBearerId := {idValue := hex2bit(tsc_EpsDefaultBearerId2ndPDN)};
    var SRB_COMMON_IND v_ReceivedAsp;
    var ProcedureTransactionIdentifier v_EpsTi;
    var B8_Type v_EsmCauseValue;

    timer t_T3412Max := f_EUTRA_SetTimerToleranceMax(eutra_CellA, nasTimer, 11160.0); // 186 minutes
    timer t_T3445Max := f_EUTRA_SetTimerToleranceMax(eutra_CellA, nasTimer, 43200.0); // 12 hours
    timer t_T3445Min := f_EUTRA_SetTimerToleranceMin(eutra_CellA, nasTimer, 43200.0); // 12 hours

    f_EUTRA_MobileInfo_SetNetworkFeatureSupport(cs_EPS_NwkFtSup_IMSVoice_EMCBS); // @sic R5s180648 sic@ emergency bearer services in S1 mode and IMS voice over PS session in S1 mode are supported

    //@siclog "Step 1" siclog@
    // The UE is switched on.
    f_EUTRA_SwitchOnUEandStartIP(eutra_CellA);

    //@siclog "Step 2" siclog@
    // Wait for 60s for the UE to enter EMM-DEREGISTERED.eCALL-INACTIVE state
    f_Delay(v_EcallDeregisteredInactive);

    //@siclog "Step 3" siclog@
    // An eCall to URI for test service is initiated by dialling the eCall Test Number in the EFFDNURI field of the USIM (tel:123456)
    f_UT_InitiateECall(UT, TEST);
      
    //@siclog "Step 3A" siclog@
    f_EUTRA_RRC_ConnectionRequest_Def(eutra_CellA, cr_EstablishmentCause_MoSignalling);
    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 3A");

    //@siclog "Step 3B-4" siclog@
    // Check: Does the UE sends an ATTACH REQUEST message with a PDN CONNECTIVITY REQUEST message to request PDN connectivity to the default PDN. EPS attach type = "combined EPS/IMSI attach"?
    v_NasInd := f_EUTRA_RRCConnectionSetupComplete_Def(eutra_CellA,
                                                       cr_508_RRCConnectionSetupComplete(tsc_RRC_TI_Def, ?),
                                                       cr_NAS_IndicationWithPiggybacking((tsc_SHT_NoSecurityProtection, tsc_SHT_IntegrityProtected),
                                                                                         cr_508_ATTACH_REQUEST(tsc_EAttachType_Combined,
                                                                                                               v_AdditionalUpdateType),
                                                                                         cr_508_PDN_CONNECTIVITY_REQUEST(tsc_PdnRequest_InitialAttach,
                                                                                                                         f_GetPdnType())));

    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 4");
    
    v_EPS_TI := v_NasInd.Pdu.PiggybackedPduList[0].Msg.pDN_CONNECTIVITY_REQUEST.procedureTransactionIdentifier;
    v_SecurityParams.NAS_SecurityCap := f_EUTRA_NB_FillNAS_SecurityCapability(v_NasInd.Pdu.Msg.aTTACH_REQUEST);
    f_EUTRA_Security_Set ( v_SecurityParams );//Saving security parameters
   
    //@siclog "Step 5-13" siclog@
    // Steps 5-13 of the generic procedure for UE registration specified in TS 36.508 subclause 4.5.2.3 are performed.
    v_APNandPCOs := f_EUTRA_InitialRegistration_Step5_13(eutra_CellA, STATE2_IDLEUPDATE, v_NasInd);

    v_PcoFromUE := v_NasInd.Pdu.PiggybackedPduList[0].Msg.pDN_CONNECTIVITY_REQUEST.protocolConfigurationOptions;
    if (isvalue(v_APNandPCOs.protocolConfigurationOptions)) {
      v_PcoFromUE := v_APNandPCOs.protocolConfigurationOptions;
    }
    v_PcoToUE := f_GetDefaultProtocolConfigOptions (v_PcoFromUE, v_PdnIndex);


    //@siclog "Step 14-15" siclog@
    // The SS transmits an ATTACH ACCEPT message with the ACTIVATE DEFAULT EPS BEARER CONTEXT REQUEST message and SS assigns value of 186 minutes for the timer T3412.
    f_EUTRA_InitialRegistration_Step14_15(eutra_CellA,
                                          NORMAL,
                                          v_NasInd.Pdu.Msg.aTTACH_REQUEST.addUpdateType,
                                          v_EPS_TI,
                                          f_CheckPCOforIPallocationViaNas (v_PcoFromUE),
                                          v_PcoToUE,
                                          v_APNandPCOs.accessPointName,
                                          v_T3412Value);// 186 minutes

    //@siclog "Step 16-17A4" siclog@
    // Steps 16-16c1 of the generic procedure for UE registration specified in TS 36.508 subclause 4.5.2.3 are performed.
    v_PdnStatus := f_EUTRA_InitialRegistration_Step16_16c1 (eutra_CellA);
        
    //@siclog "Step 17A5a1-17A5b1" siclog@
    if (v_PdnStatus != singlePDN) {     // @sic R5-227567 sic@
      if (pc_UE_supports_user_initiated_PDN_disconnect) {
        // Step 1
        f_UT_DeactivateBearer(UT, hex2int(tsc_EpsDefaultBearerId2ndPDN));
      
        // Step 9: The UE transmits PDN DISCONNECT REQUEST
        SRB.receive(car_SRB2_NasPdu_IND(eutra_CellA, cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered, cr_PDNDisconnectReq(v_LinkedEpsBearerId)))) -> value v_ReceivedAsp;
        v_EpsTi := v_ReceivedAsp.Signalling.Nas[0].Pdu.Msg.pDN_DISCONNECT_REQUEST.procedureTransactionIdentifier;
        v_EsmCauseValue := tsc_ESM_Cause36_RegularDeactivation;
      } else {
        v_EpsTi := tsc_PTI_Unassigned;
        v_EsmCauseValue := tsc_ESM_Cause26_InsufficientResources;
      }
      // Steps 10-12
      f_EUTRA_DeactivateEPS_BearerContext(eutra_CellA, tsc_EpsDefaultBearerId2ndPDN, v_EpsTi, v_EsmCauseValue); // @sic R5-171372 sic@
      f_IP_Handling_StopPDN(IP, cs_IP_StopPDN(PDN_Internet));
    }

    //@siclog "Step 18-28 Void" siclog@
    
    // Wait until registration finished and Invite received
    //REMOVED f_IMS_IPCAN_WaitForTrigger(IMS[tsc_Index_IMS1]);
    //@siclog "Step 29-31" siclog@
    // The UE performs Steps 23-25 of the generic test procedure as described in TS 36.508 sub clause 4.5A.27.3
    f_EUTRA_ActivateDedicatedEpsBearer_Speech(eutra_CellA);
    f_IMS_IPCAN_SendCoOrdMsg(IMS[tsc_Index_IMS1]);
           
    //@siclog "Step 32-36" siclog@
    // Release eCall to URI for test service using the generic procedure described in TS 34.229-1 [35] sub clause C.33. @sic R5-196019 sic@
    //IMS call has been released.
    f_IMS_IPCAN_WaitForTrigger(IMS[tsc_Index_IMS1]);
    // The SS deactivates the EPS bearer. [This is the rest of C.33]
    f_EUTRA_DeactivateEPS_BearerContext (eutra_CellA,
                                         tsc_EpsDedicatedBearerId);
 
    //@siclog "Step 37" siclog@
    // The SS releases the RRC connection.
    f_EUTRA_RRC_ConnectionRelease(eutra_CellA);
    
    // start after the end of the eCall
    // MOVED t_T3412Max.start;
    t_T3445Min.start;
    t_T3445Max.start;
    
    //@siclog "Step 38" siclog@
    // The SS waits 1 minute.
    f_Delay(60.0);

    //@siclog "Step 39-62" siclog@
    // Steps 1 to 24 of the generic test procedure for IMS MT speech call specified in TS 36.508 subclause 4.5A.7.3
    // from function f_EUTRA_GenericIMS_MT_SpeechCall(EUTRA_CellId_Type p_CellId) runs on EUTRA_PTC
    f_EUTRA_RbEst_Def(eutra_CellA); //Steps 1-8
    // Send trigger to IMS PTC
    f_IMS_IPCAN_SendCoOrdMsg(IMS[tsc_Index_IMS1]); // Steps 9 to 12 are executed in IMS PTC
    //Wait Trigger from IMS PTC: steps 1-4 have been done
    f_IMS_IPCAN_WaitForTrigger(IMS[tsc_Index_IMS1]);

    f_EUTRA_ActivateDedicatedEpsBearer_Speech(eutra_CellA, -, -, -, true);
    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 62");

    //@siclog "Step 63-66" siclog@
    // Generic test procedure for MT release of IMS call as described in annex C.33 of TS 34.229-1 [35] takes place.
    f_IMS_IPCAN_SendCoOrdMsg(IMS[tsc_Index_IMS1]);
    
    //Wait Trigger from IMS PTC: MT release has happened
    f_IMS_IPCAN_WaitForTrigger(IMS[tsc_Index_IMS1]);
    // The SS deactivates the EPS bearer. [This is the rest of C.33]
    f_EUTRA_DeactivateEPS_BearerContext (eutra_CellA,
                                         tsc_EpsDedicatedBearerId);
    
    //@siclog "Step 67" siclog@
    // The SS releases the RRC connection.
    f_EUTRA_RRC_ConnectionRelease(eutra_CellA);
    
    t_T3412Max.start;

    alt {
      //@siclog "Step 68" siclog@
      // Check: Does the UE transmit a TRACKING AREA UPDATE REQUEST message at the expiry of T3412 periodically for every 186 minutes and remains registered until T3445 expires?
      //@siclog "Step 69" siclog@
      // The SS transmits a TRACKING AREA UPDATE ACCEPT message.
      //@siclog "Step 70" siclog@
      // The UE sends TRACKING AREA UPDATE COMPLETE message.
      //@siclog "Step 71" siclog@
      // The SS releases the RRC connection.
      [] SRB.check(receive(car_SRB0_RrcPdu_IND(?, cr_RRCConnectionRequest(cr_EstablishmentCause_Any)))) {
          f_EUTRA_TrackingAreaUpdate(eutra_CellA, tsc_RRC_TI_Def, -, cr_EstablishmentCause_Any, tsc_EpsUpdate_Periodic, -, -, cs_GprsTimer_tv('5A'O, '010'B, '11111'B), cs_EPS_NwkFtSup_IMSVoice_EMCBS);
          f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 68");
          t_T3412Max.start;
          repeat;
        }
      [] t_T3412Max.timeout {
          f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 68 - TAU expected was not received in time");
        }
      [] t_T3445Min.timeout {
          v_T3445Min_elapsed := true;
          t_T3412Max.stop;
        }
      }

    v_DetachType :=  f_GetEDetachType(NORMAL);
    alt {
      //@siclog "Step 72" siclog@
      // Check: Does the UE transmit a DETACH REQUEST message at expiry of T3444?
      [] SRB.check(receive(car_SRB0_RrcPdu_IND(?, cr_RRCConnectionRequest(cr_EstablishmentCause_Any)))) {
          v_NasInd := f_EUTRA_RRC_ConnEst_DefWithNas(eutra_CellA,
                                                     tsc_RRC_TI_Def,
                                                     ?,
                                                     (cr_NAS_Indication(tsc_SHT_IntegrityProtected,
                                                                        cr_508_DETACH_REQUEST_MO(?,
                                                                                                 f_GutiParameters2MobileIdentity(omit, v_Guti1Params),
                                                                                                 cr_DetachType( '0'B, v_DetachType)))));
          t_T3445Max.stop;
          if  (v_T3445Min_elapsed) {
            f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 72");
          } else  {
            f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 72 - DETACH REQUEST expected was received too early");
          }
          //@siclog "Step 73" siclog@
          // The SS responds with DETACH ACCEPT message.
          SRB.send (cas_SRB1_NasPdu_REQ (eutra_CellA,
                                         cs_TimingInfo_Now,
                                         cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,
                                                        cs_508_DETACH_ACCEPT)));
        }
      [] t_T3445Max.timeout {
          f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 72 - DETACH REQUEST expected was not received");
        }
      }

    //@siclog "Step 74" siclog@
    // The SS releases the RRC connection.
    f_EUTRA_RRC_ConnectionRelease(eutra_CellA);
    
  }



1. [bookmark: _Toc128428303]Change 3
	Function name
	f_TC_11_3_2_IMS1()

	Reason for change
	1. Call to f_IMS_PTC_ImsInfo_GetTransportProtocol() comes to early. Transport Protocols not available yet.
2. In this test case there are no coordination triggers to inform EUTRA PTC that IMS registration has started / completed.This means that the EUTRA PTC will proceed with 17A5a1/17A5b1 (non-IMS PDN released) exactly 10s after sending of ATTACH COMPLETE, even if IMS registration is still ongoing at this time. 
3. After sending NOTIFY message, need repeat statement to repeat alt statement for further receiving of 200 OK for NOTIFY or INVITE.
4. Proper implementation is not there to check and receive 200 OK for NOTIFY.
5. Need trigger to wait for the establishment of dedicated bearers before establishment of ecall step 2-3 from C.47.
6. The procedure is not aligned with the codec requirements.
7. Trigger not needed before releasing of ecall as ecall can be release followed by ecall establishment.
8. Missing steps for MT call setup

	Summary of change
	1. Removed and Delay Call of f_IMS_PTC_ImsInfo_GetTransportProtocol() as a parameter in a_IMS_ReceiveResponse200OK() and template cs_IMS_RoutingInfo().
2. Add coordination triggers from IMS PTC so that the “waitForIMS” timer is controlled properly on EUTRA PTC.The first trigger is sent immediately since IMS registration is mandatory for this test case. 
3. Added repeat statement
4. Added statement for IMS_Client to check and receive 200 OK for NOTIFY.
5. Added “WaitForTrigger” before ecall establishment.
6. The codec requirements has been updated to AMR-WB.
7. Removed “SendCoOrdMsg” from IMS PTC before ecall release.
8. Add remaining steps for C.11 MT call setup sequence

Note - R5-231208 was submitted to RAN5#98

	TTCN module
	 IMS_ECall_IMS.ttcn

	MCC160 Comment
	



Before change
  function f_TC_11_3_2_IMS1() runs on IMS_PTC
  {
    var IMS_DATA_REQ v_IMS_REGISTER_REQ;
    var IMS_InviteRequestWithSdp_Type v_InviteRequestWithSdp;
    var SipUrl v_ContactUrl;
    var REGISTER_Request v_RegisterRequest;
    var INVITE_Request v_InviteRequest;
    var template (value) MessageHeader v_MessageHeader_Notify;
    var InternetProtocol_Type v_Protocol := f_IMS_PTC_ImsInfo_GetTransportProtocol();
    var boolean v_FinishedRegistration := false;
    var boolean v_RxdInvite := false;

    f_IMS_PTC_Init(PDN_1);

    // Don't need to wait for trigger as 1st message UL and won't be sent until PDN is in place
    
    // @siclog "Step 1 Parallel Behaviour" siclog@
    // steps 4-7 of registration procedure
    v_IMS_REGISTER_REQ := f_IMS_Register();
    v_RegisterRequest := v_IMS_REGISTER_REQ.Request.Register;
    v_ContactUrl := f_MessageHeader_GetContactSipUrl(v_RegisterRequest.msgHeader);
    
    // @siclog "Step 2 Parallel Behaviour" siclog@
    // receive INVITE in parallel to steps 8-11 of registration procedure
    alt {    // @sic R5-227567 sic@
         // Registration steps 8-9
      [] IMS_Server.check(receive(car_IMS_Subscribe_Request())) {
        f_IMS_Register_Subscribe();

         // Registration step 10
        v_MessageHeader_Notify := f_IMS_NotifyRequest_MessageHeader_A_1_6(cs_SubscriptionState_Active);  /* @sic R5-153756, R5-153986 sic@ */
        IMS_Client.send(cas_IMS_Notify_Request(cs_IMS_RoutingInfo(v_Protocol), cs_NOTIFY_Request(v_ContactUrl, v_MessageHeader_Notify,
                           cs_MessageBody_RegInfo(f_IMS_XmlMessageForNotify_Active(v_RegisterRequest)))));

        f_IMS_OtherResponse_200_MessageHeaderRX(v_MessageHeader_Notify);
      }
         // Registration step 11
      [] a_IMS_ReceiveResponse200OK(v_MessageHeader_Notify, v_Protocol){
        v_FinishedRegistration := true;
        if (not v_RxdInvite){
          repeat;
        }
      }
         // INVITE
      [] IMS_Server.check(receive(car_IMS_Invite_Request())) {
        v_InviteRequestWithSdp := f_ECall_InitialInvite_AnnexC47(TEST); /* @sic R5-220971, R5-221073 sic@ */
        f_IMS_PreliminaryPass(__FILE__, __LINE__, "INVITE Received");
        v_RxdInvite := true;
        if (not v_FinishedRegistration) {
          repeat;
        }
      }
      // allow the Register message to be repeated
      [] IMS_Server.receive(cr_IMS_DATA_REQ_Any(v_IMS_REGISTER_REQ.RoutingInfo, v_IMS_REGISTER_REQ.Request)) { repeat; }
    }
    
    // Inform EUTRA registration finished and Invite received
    f_IMS_IPCAN_SendCoOrdMsg(IPCAN);

    // Steps 2-3 of C.47
    f_ECall_CallSetup_Steps2_3 (v_InviteRequestWithSdp, { EVS_16000 }); // Split Invite into a separate function @sic R5-227567 sic@

    f_IMS_IPCAN_WaitForTrigger(IPCAN);
    v_InviteRequest := v_InviteRequestWithSdp.Invite;
    v_ContactUrl := f_MessageHeader_GetContactSipUrl(v_InviteRequest.msgHeader);
    f_IMS_CallReleaseMT(v_ContactUrl); // @sic R5-196019 sic@
    // Inform EUTRA: IMS Call released.
    f_IMS_IPCAN_SendCoOrdMsg(IPCAN);

    f_IMS_IPCAN_WaitForTrigger(IPCAN);
    v_InviteRequestWithSdp := f_IMS_MTCallSetup_BuildInviteRequestWithSdp();
    f_IMS_MTCallSetup_AnnexC11_Steps1_4(v_InviteRequestWithSdp);
    // Inform EUTRA: IMS Call established.
    f_IMS_IPCAN_SendCoOrdMsg(IPCAN);

    f_IMS_IPCAN_WaitForTrigger(IPCAN);
    v_InviteRequest := v_InviteRequestWithSdp.Invite;
    v_ContactUrl := f_MessageHeader_GetContactSipUrl(v_InviteRequest.msgHeader);
    f_IMS_CallReleaseMT(v_ContactUrl);
    // Inform EUTRA: IMS Call released.
    f_IMS_IPCAN_SendCoOrdMsg(IPCAN);

    f_IMS_IpcanReleaseWithOptionalDeregistration();
  }



After change
function f_TC_11_3_2_IMS1() runs on IMS_PTC
  {
    
    var IMS_DATA_REQ v_IMS_REGISTER_REQ;
    var IMS_InviteRequestWithSdp_Type v_InviteRequestWithSdp;
    var SipUrl v_ContactUrl;
    var REGISTER_Request v_RegisterRequest;
    var INVITE_Request v_InviteRequest;
    var template (value) MessageHeader v_MessageHeader_Notify;
    //REMOVED var InternetProtocol_Type v_Protocol := f_IMS_PTC_ImsInfo_GetTransportProtocol();
    var boolean v_FinishedRegistration := false;
    var boolean v_RxdInvite := false;
    var template (present) MessageHeader v_MessageHeader_Response;
    var IMS_Response_Type v_SessionInProgress183;
    var IMS_ResponseWithSdp_Type v_ResponseWithSdp;
   
    f_IMS_PTC_Init(PDN_1);
    f_IMS_IPCAN_SendCoOrdMsg(IPCAN);
    // Don't need to wait for trigger as 1st message UL and won't be sent until PDN is in place
    
    // @siclog "Step 1 Parallel Behaviour" siclog@
    // steps 4-7 of registration procedure
    v_IMS_REGISTER_REQ := f_IMS_Register();
    v_RegisterRequest := v_IMS_REGISTER_REQ.Request.Register;
    v_ContactUrl := f_MessageHeader_GetContactSipUrl(v_RegisterRequest.msgHeader);
    
    // @siclog "Step 2 Parallel Behaviour" siclog@
    // receive INVITE in parallel to steps 8-11 of registration procedure
    alt {    // @sic R5-227567 sic@
         // Registration steps 8-9
      [] IMS_Server.check(receive(car_IMS_Subscribe_Request())) {
        
        f_IMS_Register_Subscribe();

         // Registration step 10
        v_MessageHeader_Notify := f_IMS_NotifyRequest_MessageHeader_A_1_6(cs_SubscriptionState_Active);  /* @sic R5-153756, R5-153986 sic@ */
        IMS_Client.send(cas_IMS_Notify_Request(cs_IMS_RoutingInfo(f_IMS_PTC_ImsInfo_GetTransportProtocol()), cs_NOTIFY_Request(v_ContactUrl, v_MessageHeader_Notify,
                           cs_MessageBody_RegInfo(f_IMS_XmlMessageForNotify_Active(v_RegisterRequest)))));

        v_MessageHeader_Response := f_IMS_OtherResponse_200_MessageHeaderRX(v_MessageHeader_Notify);
        repeat;
      }
         // Registration step 11
      //MOVED [] a_IMS_ReceiveResponse200OK(v_MessageHeader_Notify, v_Protocol){
      [] IMS_Client.check(receive(car_IMS_DATA_RSP(cr_IMS_RoutingInfo(), cr_Response(c_statusLine200, v_MessageHeader_Response)))){
        a_IMS_ReceiveResponse200OK(v_MessageHeader_Notify, f_IMS_PTC_ImsInfo_GetTransportProtocol());
        v_FinishedRegistration := true;
        if (not v_RxdInvite){
         repeat;
        }
      }
         // INVITE
      [] IMS_Server.check(receive(car_IMS_Invite_Request())) {
        v_InviteRequestWithSdp := f_ECall_InitialInvite_AnnexC47(TEST); /* @sic R5-220971, R5-221073 sic@ */
        f_IMS_PreliminaryPass(__FILE__, __LINE__, "INVITE Received");
        v_RxdInvite := true;
        if (not v_FinishedRegistration) {
          repeat;
        }
      }
      // allow the Register message to be repeated
      [] IMS_Server.receive(cr_IMS_DATA_REQ_Any(v_IMS_REGISTER_REQ.RoutingInfo, v_IMS_REGISTER_REQ.Request)) { repeat; }
    }
    
    // Inform EUTRA registration finished and Invite received
    f_IMS_IPCAN_SendCoOrdMsg(IPCAN);
    f_IMS_IPCAN_WaitForTrigger(IPCAN);
    // Steps 2-3 of C.47
    f_ECall_CallSetup_Steps2_3 (v_InviteRequestWithSdp, { AMR_16000, AMR_8000 }); // Split Invite into a separate function @sic R5-227567 sic@

    //REMOVED f_IMS_IPCAN_WaitForTrigger(IPCAN);
    v_InviteRequest := v_InviteRequestWithSdp.Invite;
    v_ContactUrl := f_MessageHeader_GetContactSipUrl(v_InviteRequest.msgHeader);
    f_IMS_CallReleaseMT(v_ContactUrl); // @sic R5-196019 sic@
    // Inform EUTRA: IMS Call released.
    f_IMS_IPCAN_SendCoOrdMsg(IPCAN);
    
    f_IMS_IPCAN_WaitForTrigger(IPCAN);
    v_InviteRequestWithSdp := f_IMS_MTCallSetup_BuildInviteRequestWithSdp();
    v_InviteRequest := v_InviteRequestWithSdp.Invite;
    v_ResponseWithSdp := f_IMS_MTCallSetup_AnnexC11_Steps1_4(v_InviteRequestWithSdp);
    v_SessionInProgress183 := v_ResponseWithSdp.Response;
    f_IMS_MTCallSetup_SendPRACK_ReceiveOK(v_InviteRequest, v_SessionInProgress183.msgHeader);
    // Inform EUTRA: IMS Call established.
    f_IMS_IPCAN_SendCoOrdMsg(IPCAN);

    f_IMS_IPCAN_WaitForTrigger(IPCAN);
    // MOVED v_InviteRequest := v_InviteRequestWithSdp.Invite;
    f_IMS_MTCallSetup_AnnexC11_Steps7_13(v_InviteRequestWithSdp, v_ResponseWithSdp);
    v_ContactUrl := f_MessageHeader_GetContactSipUrl(v_InviteRequest.msgHeader);
    f_IMS_CallReleaseMT(v_ContactUrl);
    // Inform EUTRA: IMS Call released.
    f_IMS_IPCAN_SendCoOrdMsg(IPCAN);
    
    f_IMS_IpcanReleaseWithOptionalDeregistration();
  } 

1. [bookmark: _Toc128428304]Change 4
	Function name
	f_ECall_InitialInvite_AnnexC47()

	Reason for change
	1. For test tpe eCall the call is made within normal registration so the condition A7 (INVITE for creating an emergency session within an emergency registration using IMS security) should not be applied. This will cause expectred route header and others to be set incorrectly.


	Summary of change
	1. Apply condition A4 (INVITE creating a dialog) for test type eCall.


	TTCN module
	IMS_eCall_Functions.ttcn

	MCC160 Comment
	



Before change

function f_ECall_InitialInvite_AnnexC47(ECall_Type p_ECallType) runs on IMS_PTC return IMS_InviteRequestWithSdp_Type
  {
    var IMS_DATA_REQ v_IMS_DATA_REQ;
    var MIME_Message v_MIME_Message;
    var MIME_Part v_MIME_Part;
    var SDP_Message v_SdpOffer;
    var template (present) SDP_Message v_SDP_Message_Step1;
    var template (present) SipUrl v_ECallPSAP_Rx := f_SipUrl_eCall_RX(p_ECallType);
    var template (present) MessageHeader v_MessageHeader;
    var template (present) AcceptBody_List v_AcceptBodyList := { cr_AcceptBody(tsc_MimeType_EmergencyCallDataControlXML), * };    // condition A20, A21
    var template (present) SemicolonParam_List v_DispositionParams := { cr_GenericParam("handling", "optional"), * };             // condition A20, A21
    var boolean v_A8 := pc_IMS_Geolocation;
    var INVITE_A_2_1_Context_Type v_Context := A_2_1_A7;
    var charstring v_CidUrl;
    var ECall_Type v_ECallType := p_ECallType;
    
    // Step 1. Receive INVITE for an eCall
    v_MessageHeader := f_IMS_InviteRequest_MessageHeaderRX(v_ECallPSAP_Rx, -, v_Context, -, v_A8);
    v_MessageHeader.accept := cr_Accept(v_AcceptBodyList);                         // condition A20, A21, A25 @sic R5-227567 sic@
    v_MessageHeader.callInfo := cr_CallInfo(-, "EmergencyCallData.eCall.MSD");     // condition A20, A21 @sic R5-227567, R5-210324: "EmergencyCallData" instead of "emergencyCallData" sic@
    if ((v_ECallType == MANUAL) or (v_ECallType == AUTOMATIC)) { // @sic R5-220971 sic@
      v_MessageHeader.recvInfo := cr_RecvInfo_Def("EmergencyCallData.eCall.MSD");    // condition A20, A21
    }
    v_MessageHeader.contentType := cr_ContentTypeMIME;                             // condition A20, A21, A8, A25 @sic R5-217960 sic@
    v_IMS_DATA_REQ := valueof(f_IMS_INVITE_ReceiveRequest_A_2_1_Optional(v_Context, v_ECallPSAP_Rx, v_MessageHeader));
    v_MIME_Message := v_IMS_DATA_REQ.Request.Invite.messageBody.mimeMessageBody;

    // Check PIDF
    if (v_A8) {
      v_MIME_Part := f_IMS_MIME_MessageBody_GetBody(v_MIME_Message, tsc_MimeType_Pidf);
      f_IMS_DecodeAndCheckGeolocation(v_MIME_Part.body.xmlBody);
    }
    // Check SDP
    v_SDP_Message_Step1 := f_IMS_BuildSDP_EmergencyCallOffer_RX();   // Build RX SDP template using codecs AMR and AMR_WB
    v_MIME_Part := f_IMS_MIME_MessageBody_GetBody(v_MIME_Message, c_sdpApplication);
    v_SdpOffer := f_IMS_SIP_DecodeMatchAndCheckSDP(v_MIME_Part.body.sdpMessageBody, v_SDP_Message_Step1);

    // Check eCall MSD
    v_CidUrl := v_IMS_DATA_REQ.Request.Invite.msgHeader.callInfo.callInfoBody[0].url;
    f_IMS_ECall_MIME_DecodeAndCheckEmergencyCallData(v_MIME_Message, tsc_MimeType_EmergencyCallDataMSD, v_CidUrl, v_DispositionParams);

    return f_IMS_InviteRequestWithSdpMO(v_IMS_DATA_REQ, v_SdpOffer);
  }

After change

function f_ECall_InitialInvite_AnnexC47(ECall_Type p_ECallType) runs on IMS_PTC return IMS_InviteRequestWithSdp_Type
  {
    var IMS_DATA_REQ v_IMS_DATA_REQ;
    var MIME_Message v_MIME_Message;
    var MIME_Part v_MIME_Part;
    var SDP_Message v_SdpOffer;
    var template (present) SDP_Message v_SDP_Message_Step1;
    var template (present) SipUrl v_ECallPSAP_Rx := f_SipUrl_eCall_RX(p_ECallType);
    var template (present) MessageHeader v_MessageHeader;
    var template (present) AcceptBody_List v_AcceptBodyList := { cr_AcceptBody(tsc_MimeType_EmergencyCallDataControlXML), * };    // condition A20, A21
    var template (present) SemicolonParam_List v_DispositionParams := { cr_GenericParam("handling", "optional"), * };             // condition A20, A21
    var boolean v_A8 := pc_IMS_Geolocation;
    var INVITE_A_2_1_Context_Type v_Context := A_2_1_A7;
    var charstring v_CidUrl;
    var ECall_Type v_ECallType := p_ECallType;
    
    if(p_ECallType == TEST){
      v_Context := A_2_1_A4; 
    }
    
    // Step 1. Receive INVITE for an eCall
    v_MessageHeader := f_IMS_InviteRequest_MessageHeaderRX(v_ECallPSAP_Rx, -, v_Context, -, v_A8);
    v_MessageHeader.accept := cr_Accept(v_AcceptBodyList);                         // condition A20, A21, A25 @sic R5-227567 sic@
    v_MessageHeader.callInfo := cr_CallInfo(-, "EmergencyCallData.eCall.MSD");     // condition A20, A21 @sic R5-227567, R5-210324: "EmergencyCallData" instead of "emergencyCallData" sic@
    if ((v_ECallType == MANUAL) or (v_ECallType == AUTOMATIC)) { // @sic R5-220971 sic@
       v_MessageHeader.recvInfo := cr_RecvInfo_Def("EmergencyCallData.eCall.MSD");    // condition A20, A21
    }
    v_MessageHeader.contentType := cr_ContentTypeMIME;                             // condition A20, A21, A8, A25 @sic R5-217960 sic@
    v_IMS_DATA_REQ := valueof(f_IMS_INVITE_ReceiveRequest_A_2_1_Optional(v_Context, v_ECallPSAP_Rx, v_MessageHeader));
    v_MIME_Message := v_IMS_DATA_REQ.Request.Invite.messageBody.mimeMessageBody;

    // Check PIDF
    if (v_A8) {
      v_MIME_Part := f_IMS_MIME_MessageBody_GetBody(v_MIME_Message, tsc_MimeType_Pidf);
      f_IMS_DecodeAndCheckGeolocation(v_MIME_Part.body.xmlBody);
    }
    // Check SDP
    v_SDP_Message_Step1 := f_IMS_BuildSDP_EmergencyCallOffer_RX();   // Build RX SDP template using codecs AMR and AMR_WB
    v_MIME_Part := f_IMS_MIME_MessageBody_GetBody(v_MIME_Message, c_sdpApplication);
    v_SdpOffer := f_IMS_SIP_DecodeMatchAndCheckSDP(v_MIME_Part.body.sdpMessageBody, v_SDP_Message_Step1);

    // Check eCall MSD
    v_CidUrl := v_IMS_DATA_REQ.Request.Invite.msgHeader.callInfo.callInfoBody[0].url;
    f_IMS_ECall_MIME_DecodeAndCheckEmergencyCallData(v_MIME_Message, tsc_MimeType_EmergencyCallDataMSD, v_CidUrl, v_DispositionParams);

    return f_IMS_InviteRequestWithSdpMO(v_IMS_DATA_REQ, v_SdpOffer);
  }

1. [bookmark: _Toc128428305]Change 5
	Function name
	Constant tsc_ECallTestService

	Reason for change
	UE will include phone-context parameter in TEL URI built from eCall test number in EFFDNURI

	Summary of change
	Corrected the expected request-URI

	TTCN module
	IMS_eCall_Functions.ttcn

	MCC160 Comment
	





Before change
const charstring tsc_ECallTestService := "123456"; /* @sic R5-191103 sic@ 36.508 cl. 4.9.5.3
                                                        @status    APPROVED (IMS) */

After change
  const charstring tsc_ECallTestService := "123456;phone-context=test.3gpp.com"; /* @sic R5-191103 sic@ 36.508 cl. 4.9.5.3
                                                        @status    APPROVED (IMS) */

1. [bookmark: _Toc128428306]Change 6
	Function name
	f_EUTRA_TrackingAreaUpdate()

	Reason for change
	Currently the TRACKING AREA UPDATE ACCEPT message sent from function f_EUTRA_TrackingAreaUpdate() does not indicate support of emergency bearer services, this can cause an eCall-Only device to detach.

	Summary of change
	New parameter for specific NW feature support settings

	TTCN module
	EUTRA_NASSteps.ttcn

	MCC160 Comment
	




Before change
function f_EUTRA_TrackingAreaUpdate(EUTRA_CellId_Type p_CellId,
                                      RRC_TransactionIdentifier p_RRC_TI := tsc_RRC_TI_Def,
                                      EUTRA_ATTACH_Type p_ForcedAttach := NORMAL,
                                      template (present) EstablishmentCause p_EstablishmentCause := ?, //@sic R5s100539 sic@
                                      template B3_Type p_EpsUpdate_TypeValue := omit,    // @sic R5s120617 sic@
                                      UE_NetworkCap_ToBeTested p_UE_NetworkCap := DEFAULT,
                                      B1_Type p_ActiveFlag := tsc_EpsUpdate_NotActive,
                                      template (omit) GPRS_Timer p_T3412 := cs_GprsTimer_tv('5A'O, tsc_GprsTimerUnit_deact, '00000'B)) // @sic R5-165962 sic@
    runs on EUTRA_PTC
  { /* @sic R5s100707 additional MCC160 comments sic@ */
    
    /* Step 1..5: */
    f_EUTRA_TrackingAreaUpdate_Step1_5(p_CellId, p_RRC_TI, p_ForcedAttach, p_EstablishmentCause, -, p_EpsUpdate_TypeValue, p_UE_NetworkCap, p_ActiveFlag, p_T3412); // @sic R5-160777, R5-180728 sic@
    
    /* Step 6: Release RRC connection */
    f_EUTRA_RRC_ConnectionRelease(p_CellId, p_RRC_TI);
  }


After change
  function f_EUTRA_TrackingAreaUpdate(EUTRA_CellId_Type p_CellId,
                                      RRC_TransactionIdentifier p_RRC_TI := tsc_RRC_TI_Def,
                                      EUTRA_ATTACH_Type p_ForcedAttach := NORMAL,
                                      template (present) EstablishmentCause p_EstablishmentCause := ?, //@sic R5s100539 sic@
                                      template B3_Type p_EpsUpdate_TypeValue := omit,    // @sic R5s120617 sic@
                                      UE_NetworkCap_ToBeTested p_UE_NetworkCap := DEFAULT,
                                      B1_Type p_ActiveFlag := tsc_EpsUpdate_NotActive,
                                      template (omit) GPRS_Timer p_T3412 := cs_GprsTimer_tv('5A'O, tsc_GprsTimerUnit_deact, '00000'B), 
                                      template (value) EPS_NetworkFeatureSupport p_NtwFeatureSupport := cs_EPS_NwkFtSup_IMSVoice) // @sic R5-165962 sic@
    runs on EUTRA_PTC
  { /* @sic R5s100707 additional MCC160 comments sic@ */
    
    /* Step 1..5: */
    f_EUTRA_TrackingAreaUpdate_Step1_5(p_CellId, p_RRC_TI, p_ForcedAttach, p_EstablishmentCause, -, p_EpsUpdate_TypeValue, p_UE_NetworkCap, p_ActiveFlag, p_T3412, p_NtwFeatureSupport); // @sic R5-160777, R5-180728 sic@
    
    /* Step 6: Release RRC connection */
    f_EUTRA_RRC_ConnectionRelease(p_CellId, p_RRC_TI);
  }


1. [bookmark: _Toc128428307]Change 7
	 Function name
	f_EUTRA_TrackingAreaUpdate_Step1_5()

	Reason for change
	Currently the TRACKING AREA UPDATE ACCEPT message sent from function f_EUTRA_TrackingAreaUpdate() does not indicate support of emergency bearer services, this can cause an eCall-Only device to detach.


	Summary of change
	New parameter for specific NW feature support settings, pass to f_EUTRA_TrackingAreaUpdate_Accept_Complete()

	TTCN module
	EUTRA_NASSteps.ttcn

	MCC160 Comment
	


                                                       
Before change
function f_EUTRA_TrackingAreaUpdate_Step1_5(EUTRA_CellId_Type               p_CellId,
                                              RRC_TransactionIdentifier p_RRC_TI := tsc_RRC_TI_Def,
                                              EUTRA_ATTACH_Type p_ForcedAttach := NORMAL,
                                              template (present) EstablishmentCause p_EstablishmentCause := ?,
                                              template (omit) PLMN_List p_EPLMNlist := omit,
                                              template B3_Type p_EpsUpdate_TypeValue := omit,
                                              UE_NetworkCap_ToBeTested p_UE_NetworkCap := DEFAULT,    // @sic R5s120617 sic@
                                              B1_Type p_ActiveFlag := tsc_EpsUpdate_NotActive,
                                              template (omit) GPRS_Timer p_T3412 := cs_GprsTimer_tv('5A'O, tsc_GprsTimerUnit_deact, '00000'B)) // @sic R5-165962 sic@
    runs on EUTRA_PTC
  {
    var NAS_MSG_Indication_Type v_NasInd;
    var GutiParameters_Type v_GutiParams:= f_EUTRA_CellInfo_GetGuti(p_CellId);
    var B3_Type v_EpsUpdate_TypeValue := f_GetEPSTAUType(p_ForcedAttach); // @sic R5s100029, R5s110176 sic@
    var template (omit) EPS_BearerContextStatus v_EPS_ContextStatus; // @sic R5s100608 sic@
    
    /* Step 1 - 3 */
    /* The TAU Request shall be integrity-protected, but not confidentiality-protected.
       UE uses the cached security context algorithms to protect the TAU Request message.
       ( reuse of keys allocated during attach procedure )
       ( 33.401 cl. 9.1.2 Idle mode procedures in E-UTRAN ) */
    /* Check that TRACKING AREA UPDATE REQUEST message is present and conforms
       with default message ( 36.508 Table 4.7.2-27: TRACKING AREA UPDATE REQUEST ) */
    v_NasInd := f_EUTRA_TrackingAreaUpdate_Step1_3(p_CellId, // @sic R5s110632 sic@
                                                   p_RRC_TI,
                                                   p_ForcedAttach,
                                                   p_EstablishmentCause,
                                                   p_EpsUpdate_TypeValue,
                                                   p_UE_NetworkCap, // @sic R5-160777 sic@
                                                   p_ActiveFlag);
    v_EPS_ContextStatus := v_NasInd.Pdu.Msg.tRACKING_AREA_UPDATE_REQUEST.epsBearerContextStatus; // Send this back in the Accept
    
    /* Step 4 - 5: send TAU ACCEPT and receive TAU COMPLETE */
    f_EUTRA_TrackingAreaUpdate_Accept_Complete(p_CellId,
                                               v_GutiParams,
                                               v_EpsUpdate_TypeValue,
                                               v_EPS_ContextStatus,
                                               v_NasInd.Pdu.Msg.tRACKING_AREA_UPDATE_REQUEST.addUpdateType, // @sic R5s110078 sic@
                                               p_ForcedAttach,
                                               p_EPLMNlist,
                                               -,
                                               p_T3412); // @sic R5-180728 sic@
  }


After change
function f_EUTRA_TrackingAreaUpdate_Step1_5(EUTRA_CellId_Type               p_CellId,
                                              RRC_TransactionIdentifier p_RRC_TI := tsc_RRC_TI_Def,
                                              EUTRA_ATTACH_Type p_ForcedAttach := NORMAL,
                                              template (present) EstablishmentCause p_EstablishmentCause := ?,
                                              template (omit) PLMN_List p_EPLMNlist := omit,
                                              template B3_Type p_EpsUpdate_TypeValue := omit,
                                              UE_NetworkCap_ToBeTested p_UE_NetworkCap := DEFAULT,    // @sic R5s120617 sic@
                                              B1_Type p_ActiveFlag := tsc_EpsUpdate_NotActive,
                                              template (omit) GPRS_Timer p_T3412 := cs_GprsTimer_tv('5A'O, tsc_GprsTimerUnit_deact, '00000'B), 
                                              template (value) EPS_NetworkFeatureSupport p_NtwFeatureSupport := cs_EPS_NwkFtSup_IMSVoice) // @sic R5-165962 sic@
    runs on EUTRA_PTC
  {
    var NAS_MSG_Indication_Type v_NasInd;
    var GutiParameters_Type v_GutiParams:= f_EUTRA_CellInfo_GetGuti(p_CellId);
    var B3_Type v_EpsUpdate_TypeValue := f_GetEPSTAUType(p_ForcedAttach); // @sic R5s100029, R5s110176 sic@
    var template (omit) EPS_BearerContextStatus v_EPS_ContextStatus; // @sic R5s100608 sic@
    
    /* Step 1 - 3 */
    /* The TAU Request shall be integrity-protected, but not confidentiality-protected.
       UE uses the cached security context algorithms to protect the TAU Request message.
       ( reuse of keys allocated during attach procedure )
       ( 33.401 cl. 9.1.2 Idle mode procedures in E-UTRAN ) */
    /* Check that TRACKING AREA UPDATE REQUEST message is present and conforms
       with default message ( 36.508 Table 4.7.2-27: TRACKING AREA UPDATE REQUEST ) */
    v_NasInd := f_EUTRA_TrackingAreaUpdate_Step1_3(p_CellId, // @sic R5s110632 sic@
                                                   p_RRC_TI,
                                                   p_ForcedAttach,
                                                   p_EstablishmentCause,
                                                   p_EpsUpdate_TypeValue,
                                                   p_UE_NetworkCap, // @sic R5-160777 sic@
                                                   p_ActiveFlag);
    v_EPS_ContextStatus := v_NasInd.Pdu.Msg.tRACKING_AREA_UPDATE_REQUEST.epsBearerContextStatus; // Send this back in the Accept
    
    /* Step 4 - 5: send TAU ACCEPT and receive TAU COMPLETE */
    f_EUTRA_TrackingAreaUpdate_Accept_Complete(p_CellId,
                                               v_GutiParams,
                                               v_EpsUpdate_TypeValue,
                                               v_EPS_ContextStatus,
                                               v_NasInd.Pdu.Msg.tRACKING_AREA_UPDATE_REQUEST.addUpdateType, // @sic R5s110078 sic@
                                               p_ForcedAttach,
                                               p_EPLMNlist,
                                               p_NtwFeatureSupport,
                                               p_T3412); // @sic R5-180728 sic@
  }






1. [bookmark: _Toc122434493][bookmark: _Toc295288970][bookmark: _Toc325725665][bookmark: _Toc99967149][bookmark: _Toc128428308]Branches executed
[bookmark: _Toc122434494][bookmark: _Toc295288971][bookmark: _Toc325725666][bookmark: _Toc99967150]The test case was executed on EUTRA band 4 using ciphering algorithm eea1 and integrity algorithm eia1
3 [bookmark: _Toc128428309]Execution Log Files
3.1 Qualcomm SM 8450+SDX65
The Qualcomm SA515M UE passed this test case on Keysight S8704A Protocol Conformance Toolset. 
The documentation below is enclosed as evidence of the successful test case run [1]:
· Test case execution log file:
Keysight \TC_11_3_2_LOG.html
· PICS/PIXIT parameter file:
       Keysight \TC_22_3_2_PIXIT.xml
In the log file the complete test case execution as well as the PICS/PIXIT parameter settings can be seen. All TLI events are presented and message contents are fully decoded and can be verified. Preliminary verdicts and the final test case verdict can be seen in the log file.

